

Lab01 *Mathematica* Introduction

Mathematica combines the capabilities of a word processor and a sophisticated calculator. Read and follow the instructions below to obtain an introduction to *Mathematica*. You can open a section by double clicking on the cell bracket found on the right side of the notebook.

Arithmetic

The following cell is Wolfram language input to calculate $\sin^2(2\pi/3)$. Observe that *Mathematica* has names for many constants (Pi for π), capitalizes the first letter of its functions (S instead of s), uses square brackets to enclose function arguments ([] instead of ()), recognizes a space between numbers or names as multiplication, and uses ^ for exponentiation. The square appears at the end because $\sin^2(2\pi/3)$ is shorthand for $(\sin(2\pi/3))^2$. Click your mouse anywhere in the cell below and press the Shift and Enter keys together. A new cell will be created with the answer.

```
Sin[2 Pi / 3] ^ 2
```

$$\frac{3}{4}$$

Mathematica gives an exact answer unless told otherwise with the N function or use of a number with a decimal point (e.g., 3.0 indicates an approximate number whereas 3 indicates an exact number). Click your mouse anywhere in the cell below and press the Shift and Enter keys together. A new cell will be created with the answers.

```
N[Sin[2 Pi / 3] ^ 2]
```

```
Sin[2. Pi / 3] ^ 2
```

0.75

0.75

Text and input can be formatted. The input in the next cell was obtained by using the Control and / keys together to obtain the fraction, the three keys `ESC p ESC` to obtain the π symbol, and the Control and ^ keys together to obtain the superscript. Alternatively, Palettes | Basic Math Assistant allows a user to use clicks to format. Click your mouse anywhere in the cell below, press the Shift and Enter keys together, and examine the resulting output.

```
Sin[2π / 3]2
```

$$\frac{3}{4}$$

Exercise. Use *Mathematica* to calculate an exact value and an approximate value of $\sqrt{e^3 + \tan(\pi/6)}$ where e is the exponential constant.

```
 $\sqrt{e^3 + \tan(\pi/6)}$   
N[%]
```

$$\sqrt{\frac{1}{\sqrt{3}} + e^3}$$

4.54564

Free-Form Input

Many of the syntactical nuances of the Wolfram language can be avoided by using free form input. Hopefully the intent of the following free-form input is self-explanatory. Note the syntax of the actual *Mathematica* command that is executed.

In a new input cell, enter `=factor x^2 - 3x`

```
factor x^2-3x
Factor[-3 * x + x^2]
```

$(-3 + x) x$


In a new input cell, enter `=solve y = x^2 - 3x for x`

```
solve y=x^2-3x for x
Solve[y == -3 * x + x^2, x, MaxExtraConditions -> Automatic]
```

$\left\{ \left\{ x \rightarrow \frac{1}{2} \left(3 - \sqrt{9 + 4 y} \right) \right\}, \left\{ x \rightarrow \frac{1}{2} \left(3 + \sqrt{9 + 4 y} \right) \right\} \right\}$

In a new input cell, enter `=graph y = x^2 - 3x`

```
graph y=x^2-3x
Plots (1 of 2)
Plot[-3 * x + x^2, {x, -0.3, 3.3}]
```




In a new input cell, enter `=minimize y = x^2 - 3x`

```
minimize y=x^2-3x
Minimize[x^2 - 3 * x, {x}]
```

$\left\{ -\frac{9}{4}, \left\{ x \rightarrow \frac{3}{2} \right\} \right\}$

In a new input cell, enter `=whopper+shake`

whopper+shake » 

↳ Total nutrition facts

serving sizes (total: 522 g)
 Burger King Whopper: 1 whopper (292 g)
 milkshake: 240 mL (230 g)

total calories 925

	% daily value*
total fat 44 g	68%
saturated fat 17 g	85%
trans fat 1 g	
cholesterol 113 mg	38%
sodium 1 g	51%
total carbohydrates 97 g	32%
dietary fiber 5 g	20%
sugar 52 g	
protein 39 g	78%

vitamin A 6%	vitamin C 1%
calcium 29%	iron 3%
vitamin E 1%	thiamin 6%
riboflavin 41%	niacin 2%
vitamin B6 5%	vitamin B12 17%
folate 2%	phosphorus 25%
magnesium 8%	zinc 7%

*percent daily values are based on a 2000 calorie diet


In a new input cell, enter `=temperature goshen indiana on 7/1/2016`

temperature goshen indiana on 7/1/2016 » 

```
AirTemperatureData[ Goshen (city) , DateObject[{2016, 7, 1}]]
```

Interval[{55.04, 71.96}] °F

In a new input cell, enter `=u.s. deficit / u.s. gnp`

u.s. deficit / u.s. gnp 

```
United States (country) [EntityProperty["Country", "FederalDeficit", {"Frequency" -> "Annual"}]] /  
United States (country) [ GNP | nominal; quarterly; seasonally adjusted ]
```

0.0238595

Free-form input is a great way to ask *Mathematica* to do something without much concern about syntax. Unfortunately, there are many computational capabilities that can only be accessed by using the Wolfram language, such as assigning names to expressions and defining your own functions.

Names

In mathematics, we speak about the *expression* $x^2 - 3x$, the *equation* $y = x^2 - 3x$, and the *function* f defined by the rule $f(x) = x^2 - 3x$. We often call x and y *variables*. We can factor, simplify, or expand an expression. We can evaluate an expression, equation, or function at a particular value. We can also obtain the graph of an equation or function.

In *Mathematica*, the concepts of expression, equation, and function remain the same, although the syntax is somewhat different for equations and functions. *Names* in *Mathematica* are often referred to as *variables*; however, the concept of a *Mathematica name* is different from the concept of a mathematical *variable*. A *Mathematica name* is an abbreviation for an expression, in computerese, storage for an expression.

In *Mathematica*, a name is a string of letters, numbers, and special characters. Names must not begin with a number, and by convention, names beginning with an upper case letter are reserved for *Mathematica* functions and constants. Click your mouse anywhere in the cell below, press the Shift and Enter keys together, and examine the resulting output.

```
thisIsNameNumber1
```

```
thisIsNameNumber1
```

`N` is the name for the *Mathematica* function that finds the numerical approximation for an expression, and `Pi` and π are the *Mathematica* names for the ratio of the circumference to the diameter of any Euclidean circle. Click your mouse anywhere in the cell below, press the Shift and Enter keys together, and examine the resulting output.

```
N[Pi]
```

```
N[ $\pi$ , 50]
```

```
3.14159
```

```
3.1415926535897932384626433832795028841971693993751
```

Names can be assigned values. Notice the color of the name “seven” is blue before executing and black after executing the following cell.

```
seven = 7
```

```
7
```

Names can be used in expressions. Examine the output after executing the following cell.

```
thisIsNameNumber1 + seven + 2
```

```
9 + thisIsNameNumber1
```

Expressions

In *Mathematica*, the equals sign (=) means assignment. When you execute the following command, the name `y` is assigned the expression $x^2 - 3x$.

```
y = x^2 - 3x
```

```
- 3x + x^2
```

This means that when you execute the next command, the output is the expression.

```
y
```

```
- 3x + x^2
```

When you execute the next command, the output echos the input because the name `x` has not been assigned

an expression.

`x`

`x`

When we assign an expression to `x`, the name `y` is updated, too.

`x = 5`

`y`

5

10

We can clear the assignment to `x`, and `y` returns to its former state.

`Clear[x]`

`y`

$-3x + x^2$


We can use `y` any place we might have used the expression that was assigned to `y`. The following commands `Factor`, `Plot`, and `Minimize` the expression.

`Factor[y]`

`Plot[y, {x, -2, 5}]`

`Minimize[y, x]`

$(-3 + x)x$



$\left\{-\frac{9}{4}, \left\{x \rightarrow \frac{3}{2}\right\}\right\}$

If we want to evaluate `y` for a certain value of `x` without assigning a value to `x`, the replacement operator (`/.`) can be used.

`y /. x → 5`

`x`

10

`x`

Equations

If you are asked to solve the equation $y = x^2 - 3x$ for `x`, you may be tempted to execute the following.

```
Solve[y = x^2 - 3 x, x]
```

Solve::naqs : $-3x + x^2$ is not a quantified system of equations and inequalities. >>

```
Solve[-3 x + x^2, x]
```

The output is somewhat cryptic. The problem is that the command assigns the expression $x^2 - 3x$ to the name y , and that expression is not an equation.

Since *Mathematica* uses the equals sign for assignment, a different symbol must be used for the mathematics equal sign. *Mathematica* uses two equals signs ($==$) to represent the mathematics equal sign. The correct solve command is the following. Note that we first cleared the name y because otherwise we would have asked *Mathematica* to solve the equation $x^2 - 3x = x^2 - 3x$.

```
Clear[y]
```

```
Solve[y == x^2 - 3 x, x]
```

$$\left\{ \left\{ x \rightarrow \frac{1}{2} \left(3 - \sqrt{9 + 4y} \right) \right\}, \left\{ x \rightarrow \frac{1}{2} \left(3 + \sqrt{9 + 4y} \right) \right\} \right\}$$

The output is a list of lists of replacement rules corresponding to the solutions. This form for the set of solutions will make more sense if we are asked to solve the system of two equations $x^2 + y^2 = 45$ and $y = 2x$. In comparison with the earlier solve command, we replace the single equation with a list of equations and we replace the single variable with a list of variables.


```
Solve[{x^2 + y^2 == 45, y == 2 x}, {x, y}]
```

```
{{x -> -3, y -> -6}, {x -> 3, y -> 6}}
```

The output tells us that one solution is $(x, y) = (-3, -6)$ and a second solution is $(x, y) = (3, 6)$.


The `Plot` function expects an expression. Hence, providing an equation does not work.

```
Plot[y == x^2 - 3 x, {x, -2, 5}]
```



On the other hand, the `ContourPlot` function expects an equation involving two variables.

```
ContourPlot[y == x^2 - 3 x, {x, -2, 5}, {y, -3, 10}]
```



Functions

To define the function f defined by the rule $f(x) = x^2 - 3x$ in *Mathematica*,

1. use square brackets in place of parentheses,
2. use the delayed assignment operator (`:=`) instead of the equals sign, and
3. place an underscore character (`_`) after the independent variable name on the left, but not right, hand side.

```
f[x_] := x^2 - 3 x
```

This command has effectively made an infinite number of assignments of the form

$$f[\textit{anything}] = \textit{anything}^2 - 3 \textit{anything}.$$

Thus, to evaluate $f(5)$, execute the following command.

```
f[5]
```

```
10
```

You can evaluate with respect to variables and expressions.

```
f[t]
```

```
f[x]
```

```
Expand[f[x + 7]]
```

```
-3 t + t^2
```

```
-3 x + x^2
```

```
28 + 11 x + x^2
```

Note that the result of such an evaluation is an expression, not a function. Hence, the second command in the following cell is meaningless.

```

h = f[x]
h[5]
-3 x + x^2
(-3 x + x^2) [5]

```

Since we have not assigned an expression to the name `f`, the output echos the input for the following command.

```

f
f

```

We can see the rule associated with a name defined to be a function by prepending a question mark.

```
? f
```

```
Global`f
```

```
f[x_] := x^2 - 3 x
```

Below are two ways to define a piecewise defined function. The second, more readable, format was obtained using the Advanced tab of the Calculator section of the Basic Math Assistant or Classroom Assistant palettes.

```
g[x_] := Piecewise[{{x^2, x < 0}, {x + 1, x >= 0}}]
```

```
g[x_] := { x^2    x < 0
          { x + 1  x >= 0
```

With two functions defined, it is easy to find compositions.

```
f[g[x]]
```

```
Simplify[%]
```

$$-3 \left(\begin{cases} x^2 & x < 0 \\ 1 + x & x \geq 0 \\ 0 & \text{True} \end{cases} \right) + \left(\begin{cases} x^2 & x < 0 \\ 1 + x & x \geq 0 \\ 0 & \text{True} \end{cases} \right)^2$$

$$\begin{cases} (-2 + x)(1 + x) & x \geq 0 \\ x^2(-3 + x^2) & \text{True} \end{cases}$$

```
g[f[x]]
```

```
Simplify[%]
```

$$\begin{cases} (-3x + x^2)^2 & -3x + x^2 < 0 \\ 1 - 3x + x^2 & -3x + x^2 \geq 0 \\ 0 & \text{True} \end{cases}$$

$$\begin{cases} (-3x + x^2)^2 & 0 < x < 3 \\ 1 - 3x + x^2 & \text{True} \end{cases}$$

Comparing a Model to Data

Name some bivariate data. Here `cdata` are the waist high circumferences in inches and `vdata` are the usable lumber in board-feet of eight Ponderosa pine trees.


```

cdata = {5.4, 6.4, 7.3, 8.9, 10.2, 11.5, 12.1, 13.1};
vdata = {190, 320, 570, 1130, 1230, 1920, 2520, 2940};
data = Transpose[{cdata, vdata}]
{{5.4, 190}, {6.4, 320}, {7.3, 570}, {8.9, 1130},
 {10.2, 1230}, {11.5, 1920}, {12.1, 2520}, {13.1, 2940}}

```


Generate a table of the data.

```
Grid[Transpose[Prepend[data, {"c", "v"}]], Frame → All]
```

c	5.4	6.4	7.3	8.9	10.2	11.5	12.1	13.1
v	190	320	570	1130	1230	1920	2520	2940

Display and name a scatter diagram of the data.

```
dataplot = ListPlot[data]
```



The data appears to be somewhat in the shape of a parabola; hence, $v = ax^2 + bx + c$ would be a reasonable model for volume v as a function of circumference x if the parameters a , b , and c are chosen appropriately. Find parameter values so that the model passes through the first, fifth, and last data points.

```

Solve[{190 == a 5.4^2 + b 5.4 + c, 1230 == a 10.2^2 + b 10.2 + c, 2940 == a 13.1^2 + b 13.1 + c}, {a, b, c}]
{{a → 48.4401, b → -538.998, c → 1688.08}}

```


Name a model of the relationship between the volume v and circumference x using the above values for the parameters.

```
model = 48.4 x^2 - 539 x + 1688
```

```
1688 - 539 x + 48.4 x^2
```


Display and name a plot of the model.

```
modelplot = Plot[model, {x, 5, 14}, PlotStyle → Red]
```



Display the model and data plots together with the axes replaced by a frame and explanatory labels included.

```
Show[{modelplot, dataplot}, Axes → False, Frame → True,
  FrameLabel → {"Circumference (in)", "Volume (board feet)"}, PlotLabel → "Ponderosa Pine"]
```



Calculate and name the volumes predicted by the model for the data set circumferences.

```
vpredicted = model /. x → cdata
```

```
{188.744, 220.864, 332.536, 724.664, 1225.74, 1890.4, 2252.34, 2933.02}
```

Calculate and name the absolute errors in the model predictions with the data.

```
errors = Abs[vpredicted - vdata]
```

```
{1.256, 99.136, 237.464, 405.336, 4.264, 29.6, 267.656, 6.976}
```

Calculate the mean, root mean square, and maximum absolute errors.

```
Mean[errors]
```

```
RootMeanSquare[errors]
```

```
Max[errors]
```

```
131.461
```

```
194.647
```

```
405.336
```


Find the quadratic model that minimizes the root mean square error, plot the data and model together, and calculate error summaries.

```

model = a x2 + b x + c;
parameters = {a, b, c};
bestparameters = FindFit[data, model, parameters, x]
bestmodel = model /. bestparameters
modelplot = Plot[bestmodel, {x, 5, 14}, PlotStyle → Red];
Show[{modelplot, dataplot}, Axes → False, Frame → True,
  FrameLabel → {"Circumference (in)", "Volume (board feet)"}, PlotLabel → "Ponderosa Pine"]
vpredicted = bestmodel /. x → cdata;
errors = Abs[vpredicted - vdata];
{Mean[errors], RootMeanSquare[errors], Max[errors]}
{a → 34.387, b → -282.833, c → 746.95}

```

$$746.95 - 282.833 x + 34.387 x^2$$



```
{98.1077, 122.773, 209.682}
```

Summary

By thoughtfully working through this notebook, you have reviewed several mathematical concepts (arithmetic, expression, equation, function, graph, factor, solving equations, and modeling data) and how to use *Mathematica* to carry out a variety of computations. Save this notebook for future reference.

Your next step is to solve the Record Time Problem and write a Report. The report should be contained in a new *Mathematica* notebook; however, you may want to copy some of the commands from this notebook to paste into your report notebook.